
برخي تكنيك​هاي برنامه​سازي و عناصر برنامه

17-1-  توابع يا رويه​هاي بازگشتي
بنابر تعريف، يك مسئله بازگشتي است، اگر براي حل آن نياز به حل زيرمسئله​اي دقيقا مشابه مسئله اصلي در ابعاد كوچكتر داشته باشيم. ممكن است حل يك مسئله با استفاده از خودش، كمي عجيب به نظر برسد، اما در بسياري از موارد، بسيار ساده​تر است. با اينگونه تعريف در رياضي آشنا هستيم. يك نمونه از مسئله ساده بازگشتي، فاكتوريل است. براي فاكتوريل گرفتن از يك عدد، مي​توان از فاكتوريل اعداد كوچكتر از آن عدد استفاده كرد:
N! = n * (n-1)!

               (n-1)! := (n-1) * (n-2)!

                                          (n-2)! = (n-2) * (n-3)!
البته بايد شرطي وجود داشته باشد كه يك مسئله بازگشتي در آن خاتمه پذيرد. در غير اين صورت مسئله هيچگاه پايان نمي​پذيرد. شرط بازگشتي در مسئله فاكتوريل به صورت زير است:

1! = 1 , 0! = 1

بنابراين هر مسئله بازگشتي يك نقطه پايان دارد.

نمونه ديگري از مسئله بازگشتي، تركيب m از n است:

C(n,m) = C(n-1 , m) + C(n-1 , m-1) 

شرط پايان براي اين مسئله در حالتهاي n=m يا n=0 است، كه در اين حالتها تركيب برابر با يك است.

مي​خواهيم الگوريتم فاكتوريل را بار ديگر بنويسيم. اين كار را مي​خواهيم بدون استفاده از حلقه انجام دهيم. ابتدا سعي مي​كينم تا قاعده كار را بيابيم. مي​دانيم كه فاكتوريل n  برابر است با 1*2*3..*n
اما مي دانيم:                          fact ( n ) =  n * (n-1)*(n-2)*….*3*2*1

بنابراين:                                                 fact ( n ) = n * fact(n-1) ;

از همين قاعده براي نوشتن الگوريتم فاكتوريل استفاده مي​كنيم. ممكن است كه بگوييد چطور fact(n-1) را محاسبه كنيم؟ خوب ساده است. جاي n-1، n قرار مي​دهيم و دوباره محاسبه مي​كنيم. اين همان تعريف بازگشتي است.

Function fact ( n : integer ) : int64 ;

  Begin 

      result := n * fact ( n-1 ) ;

  End ;
به همين سادگي! اما واقعا چه اتفاقي در روند اجراي برنامه پيش مي​آيد؟ هر بار تابع خودش را با مقدار كمتر صدا مي زند، اما هيچ​گاه متوقف نمي​شود. بنابراين در يك حلقه بي پايان مي​ماند. از اين رو بايد براي آن پاياني نيز در نظر بگيريم. در اين صورت برنامه به شكل زير در خواهد آمد:
Function fact ( n : integer ) : int64 ;

  Begin 

      if n > 1 then  result := n * fact ( n-1 ) 

                   else   result := 1 ;
 End ;
اين تابع آنقدر خودش را صدا مي زند تا به يك برسد. به اين تابع يك تابع بازگشتي يا recursive گفته مي​شود.
اين نوع از توابع يا رويه​ها، براي موضوعاتي كه ماهيت تكراري دارند كاربرد بسياري دارند. معمولا در صورتي كه مسئله ماهيت بازگشتي داشته باشد، و برنامه​نويس هم درست بتواند مسئله را در ذهن خودش به شكل بازگشتي مجسم كند، نوشتن برنامه بسيار ساده و حتي از برنامه​هاي غير بازگشتي ساده​تر خواهد بود. برنامه​سازي بازگشتي يكي از امكانات بسيار قوي در زبانهاي برنامه​سازي است.

اما واي به حال برنامه​نويسي كه بخواهد اين نوع از برنامه​ها را به شكل غيربازگشتي تفسير و دنبال كند! چرا كه اين برنامه​ها ماهيتا قابل دنبال كردن ياtrace  نيستند. بلكه فقط با بررسي و درك منطق آنها مي​توان عمل آنها را كنترل نمود. مهمترين چيز براي نوشتن توابع بازگشتي، قدرت تجسم برنامه​نويس است. و اين امر بجز با تمرينهاي مكرر ذهني امكان ندارد.

ممكن است كه اين سؤال مطرح شود كه بهتر است مسئله را به صورت بازگشتي نوشت، يا بصورت تكرار؟ اين سؤال جواب روشني ندارد. انتخاب الگوريتم، به ماهيت مسئله بستگي دارد. گاهي الگوريتم تكرار بهتر است و گاهي بازگشتي. البته دنبال​كردن برنامه​هاي بازگشتي مشكل​تر از برنامه​هاي غيربازگشتي است.
اجراي برنامه​هاي بازگشتي نسبت به برنامه​هاي غيربازگشتي داراي كارائي (سرعت و مصرف حافظه كم) مناسبي نيست. بنابراين وقتي بايد از الگوريتمهاي بازگشتي استفاده كرد كه ماهيت برنامه به گونه​اي باشد كه درك برنامه را بسيار ساده​تر از درك برنامه غيربازگشتي كند و در آن تكه از برنامه، كارائي آنچنان اهميت نداشته باشد (نسبت به قابليت درك برنامه).

مثال 18-1

برنامه​اي بنويسيد كه يك خط از متن را كه طول آن نامشخص است، از ورودي بخواند تا اينكه به پايان خط برسد. آنگاه خط متن را به ترتيب عكس خوانده​شدن چاپ كند.
در اين برنامه از تابعي با نام EOLN استفاده خواهيم كرد. اين تابع بررسي مي​كند كه آيا خط به پايان رسيده است يا خير. بنابراين خروجي اين تابع يك عبارت منطقي يا بولين است. اين برنامه را در محيط كنسول مي​نويسيم.
Program BackWard

Procedure flipit ;

Var

  C : char ;

Begin
  Read (c) ;

  If not eoln then

      Flipit ;

  Write(k) ;

End ;

Begin
  Writeln(‘Enter a line of text’) ;

  Flipit ;

            End ;

تمرين

برنامه​هاي زير را به صورت بازگشتي بنويسيد.

- محاسبه مجموع n عدد

- جستجو در يك آرايه به روش باينري

17-2- متغيرهاي بدون نوع (Variant variable)

متغيرهايي وجود دارند كه در هنگام تعريف، نوعشان معلوم نيست و هنگامي كه مقداري در آنها قرار مي​گيرد، نوعشان برحسب نوع آن مقدار، معلوم مي​شود. به اين متغيرها، متغيرهاي بدون نوع گوييم. بنابراين هر نوع مقداري و عبارتي را مي​توان در چنين متغيرهايي ريخت. براي تعريف اين متغيرها از كلمه variant استفاده مي​شود. مثال زير استفاده از چنين متغيرهايي را نشان مي​دهد:
 var

    V1, V2, V3, V4, V5: Variant;

    I: Integer;

    D: Double;

    S:  string ;

  Begin
    V1 := 1; { integer value }

    V2 := 1234.5678; { real value }

    V3 := 'Hello world!'; { string value }

    V4 := '1000'; { string value }

    V5 := V1 + V2 + V4; { real value 2235.5678}

    I := V1; { I = 1 (integer value) }

    D := V2; { D = 1234.5678 (real value) }

    S := V3; { S = 'Hello world!' (string value) }

    I := V4; { I = 1000 (integer value) }

    S := V5; { S = '2235.5678' (string value) }

 End ;

البته استفاده از متغير بدون نوع، باعث كاهش كارائي برنامه مي​شود و بجز در مواردي كه واقعا مجبور به استفاده از آن نشويم، نبايد از آن استفاده كنيم. (هر چند كه برنامه​نويسان ناشي براي فرار از ايجاد يك برنامه ساخت​يافته از آن استفاده مي​كنند). همچنين استفاده از اين​گونه از متغيرها بايد  با در نظر گرفتن نكات مربوط به تبديل نوع​ها  به هم باشد كه در راهنماي دلفي بيان شده است.

17-3- آرايه​هاي پويا

آرايه​هاي پويا اندازه و طول ثابتي ندارند. ممكن است كه اين سؤال ايجاد شود كه پس چگونه حافظه به اين نوع آرايه​ها اختصاص داده مي​شود؟ در جواب بايد گفت كه، به حافظه يك آرايه پويا در هنگام اضافه شدن يك مقدار به طول آن، تخصيصي دوباره داده مي​شود (Reallocation). اين آرايه​ها به شكل زير تعريف مي​گردند:

Var  MyFlexibleArray : array of Real ;
و براي استفاده از اين آرايه و تعيين اندازه آن، از تابع Setlength استفاده مي​شود. براي مثال دستور زير فضايي با 20 خانه براي اين آرايه در نظر مي​گيرد:

SetLength(MyFlexibleArray, 20) ;
آرايه​هاي پويا هميشه انديس عدد صحيح دارند و از صفر شروع مي​گردند. در تكه برنامه​هاي زير از آرايه پويا استفاده شده است:
var

   A :  array   of  string;

   I : Integer;

  Begin
    SetLength(A, 10);

    for  I := Low(A)  to  High(A)  do

      Begin
        A[i] := inttostr ( I )  ;

      End ;

  End ;

Function  Max(A:  array   of  Real; N: Integer): Real;

 var

    X: Real;

    I: Integer;

  Begin
    X := A[0];

     for  I := 1  to  N - 1  do

       if  X < A[I]  then  X := A[I];

    Max := X;

  End ;
آرايه​هاي پويا مي​توانند به شكل چندبعدي نيز تعريف شوند، مثلا:

var

   A :  array   of   array   of  string;

    I, J : Integer;

  Begin
    SetLength(A, 10);

     for  I := Low(A)  to  High(A)  do

     Begin
      SetLength(A[I], I);

       for  J := Low(A[I])  to  High(A[I])  do

        A[I,J] := IntToStr(I) + ',' + IntToStr(J) + ' ';

     End ;

            End ;
17-3-1 آرايه​هاي پويا به عنوان پارامتر
در اين نوع پارامترها، آرايه​هاي با اندازه متفاوت به تابع يا رويه ارسال شوند. براي تعريف يك رويه با چنين پارامتري به صورت زير عمل مي​كنيم:
Function Find(A: array of Char): Integer ;
مثال بالا تابعي تعريف مي​كند كه هر آرايه از كاراكتر را كه شامل آرايه​هاي پويا است، در بر مي​گيرد. براي اعلان پارامترهايي كه بايد آرايه​هاي پويا باشند، بايستي يك نوع مشخصي تعريف شود:

type TDynamicCharArray = array of Char ;

Function Find(A: TDynamicCharArray): Integer ;
مثال​هايي از اين نوع پارامترها در زير آمده است:

procedure  Clear( var  A:  array   of  Real);

 var

    I: Integer;

  Begin
     for  I := 0  to  High(A)  do  A[I] := 0;

  End ;

Function  Sum( const  A:  array   of  Real): Real;

 var

    I: Integer;

    S: Real;

  Begin
    S := 0;

     for  I := 0  to  High(A)  do  S := S + A[I];

    Sum := S;

  End ;

چند نكته در استفاده از آرايه​هاي پويا به عنوان پارامتر:
اين پارامترها هميشه از مبناي صفر آغاز مي​شوند. يعني عنصر اول، صفر، عنصر دوم، يك و الي​آخر. توابع Low و High بر روي اين نوع پارامترها، به ترتيب صفر و طول منهاي يك را برمي​گردانند. تابع SizeOf اندازه آرايه واقعي فرستاده شده به رويه را برمي​گرداند.
اين پارامترها تنها از طريق عناصر، قابل دسترسي هستند. انتصابات روي اين نوع پارامترها قابل انجام نيست.
اين پارامترها مي​توانند به توابع يا رويه​هاي ديگر تنها از طريق پارامترهاي آرايه پويا و يا پارامترهاي بدون نوع. همچنين آنها نمي​توانند ازطريق تابع SetLength ارسال گردند.
بجاي يك آرايه، مي​توان يك متغير از نوع آرايه پويا ارسال نمود. اين متغير همانند يك آرايه با طول يك رفتار خواهد نمود.
17-3-2- سازنده​هاي آرايه پويا

اگر رويه​اي با پارامتر آرايه پويا داشته باشيم، مانند مثال زير:

procedure  Add(A:  array   of  Integer); 
در اين صورت مي​توانيم رويه Add را با اين عبارت فراخواني كنيم.
Add([5, 7, I, I + J]); 
در اين صورت، اين دستور معادل دستور زير است:
Var  Temp:  array [0..3]  of  Integer; 

 ... 

 Temp[0] := 5;  Temp[1] := 7;  Temp[2] := I;  Temp[3] := I + J; 

 Add(Temp); 
تمرين
برنامه​اي بنويسيد كه يك آرايه با هر اندازه را مرتب كند.
برنامه​اي بنويسيد كه در يك آرايه جستجو كند و محل عنصري را بيابد.

17-4- Absolute

اگر بخواهيد دو متغير را در يك نقط يكسان از حافظه تعريف كنيد، چه بايد بكنيد؟ دستور Absolute چنين امكاني را فراهم مي​كند. ساده​ترين كاربرد اين دستور زماني است كه بخواهيد تمام يا بخش اول يك متغير از يك نوع را به شكل متغير ديگري با نوع ديگر مشاهده كرده و آن را بخوانيد يا تغيير دهيد. تكه برنامه زير، يك كاربرد از Absolute را نشان مي​دهد:
Var  

  s       : string[120] ; 

  slen  : byte absolute s ; 

Begin 

    s := ‘alireza’ ; 

    slen := 3 ;

                   // s = ‘ali’ 

End ;
17-5- پارامترهاي بدون نوع

اگر بخواهيد به يك روال، پارامتري را بفرستيد كه نوع آن از قبل معلوم نيست، چه بايد بكنيد؟ تكه برنامه زير چگونگي تعريف و استفاده از پارامترهاي بدون نوع را نشان مي​دهد:
Function  Equal( var  Source, Dest; Size: Integer): Boolean;

 type

    TBytes =  array [0..MaxInt - 1]  of  Byte;

  var

    N: Integer;

  Begin
    N := 0;

     while  (N < Size)  and  (TBytes(Dest)[N] = TBytes(Source)[N])  do

      Inc(N);

    Equal := N = Size;

  End ;

  ...

   i,j : integer ;

    ....

   if equal ( i,j,4 ) then ... ;

   s1,s2 : string ;

   ....

   if equal ( s1,s2,length (s1) ) then ... ;

تمرين

تابعي بنويسيد كه دو متغير از هر نوع را با هم Swap كند.

17-6- پارامترهاي با مقدار پيش​فرض

در صورتي كه بخواهيم يك مقدار پيش​فرض را به يك روال ارسال كنيم، از اين پارامتر استفاده مي​كنيم. مثال زير چگونگي اين كار را نشان مي​دهد:

procedure  FillArray(A:  array   of  Integer; Value: Integer = 0);

بنابراين فراخواني​هاي زير با هم معدلند:

FillArray(MyArray);

FillArray(MyArray, 0);

17-7- بارگيري مكرر (Overloading
)

در صورتي كه بخواهيم يك روال با يك نام واحد، ولي شكل تعريف مختلف و امكان تعريف پارامترهاي مختلف و متفاوت را تعريف كنيم، از بارگيري مكرر استفاده مي​كنيم. مثال زير يك نمونه از كاربرد اين امكان را نشان مي​دهد:
Function  Divide(X, Y: Real): Real;  overload ;

 Begin
    Result := X/Y;

  End ;

 Function  Divide(X, Y: Integer): Integer;  overload ;

 Begin
    Result := X  div  Y;

  End ;
با اين كار، دو تابع، كه هر دوي آنها Divide نام دارند، تعريف مي​شوند. البته نوع پارامترهايي كه اين دو تايع مي​گيرند، باهم متفاوت است. هنگامي كه تابعي با نام Divide در برنامه فراخواني شود، بر حسب اين كه پارامترهاي آن از چه نوعي باشند، تابع با همان پارامتر فراخواني مي​شود. مثلا فراخواني Divide(6.0,3.0)، تابع اولي را فراخواني مي​كند، چرا كه آرگومان​هاي آن هر دو عد حقيقي هستند.

17-8- دستور Goto
اين دستور، كنترل برنامه را به صورت غيرشرطي، به بخشهاي مختلف برنامه مي​فرستد. براي استفاده از اين دستور نياز به برچسبهايي داريم. برچسبها، نشان​دهنده مكانهايي از برنامه هستند كه توسط دستور Goto كنترل برنامه به آن مكانها انتقال پيدا مي​كند. براي تعريف يك برچسب، بعد از كلمه Label، نام برچسب را مي​نويسيم و جايي كه لازم است تا كنترل برنامه به آن مكان انتقال پيدا كند، با نام برچسب در ابتداي خط آن مكان و يك ":" بعد از آن مشخص مي​كنيم. برنامه زير استفاده از اين پرش را نشان مي​دهد:

procedure  FindFirstAnswer;

 var  X, Y, Z, Count: Integer;

  label  FoundAnAnswer;

  Begin
    Count := SomeConstant;

     for  X := 1  to  Count  do

       for  Y := 1  to  Count  do

         for  Z := 1  to  Count  do

           if  ... { some condition holds on X, Y, and Z }  then

             goto  FoundAnAnswer;

    ... {code to execute if no answer is found }

    Exit;

    FoundAnAnswer:

      ... { code to execute when an answer is found }

 End ;
توجه مهم

دستور Goto براي برنامه​نويسي ساخت يافته مي​تواند بسيار خطرناك باشد. اين دستور پرش مي​تواند باعث درهم و برهم شدن ساختمان برنامه شود؛ به نحوي كه منطق برنامه كاملا نامشخص شود. برنامه​نويسان ناشي براي فرار از قسمتهايي كه در آن گير مي​كنند، از اين دستور استفاده مي​كنند. در چنين مواردي استفاده از اين دستور يك سم مهلك است نه يك دارو. اغلب برنامه​هايي كه از اين دستور استفاده مي​كنند خوانايي كمي دارند و به همين ترتيب اشكال​زدايي در آنها بسيار مشكل است.
ما استفاده از اين دستور را پيشنهاد نمي​كنيم، ولي اگر هم بخواهيم از آن استفاده كنيم، بايد به نكات زير توجه داشته باشيم. 
17-8-1- نكات مهم در پرش
تمام پرش​ها بايد در يك تكه كوچك از برنامه و فقط در درون يك زيربرنامه انجام شود.
همه پرش​ها بايد يا رو به جلو و يا رو به عقب انجام شود.

به چيزهائي كه از روي آنها مي​پريد توجه داشته باشيد. بدانيد به كجا مي​پريد، و از روي چه چيزهائي مي​پريد. دستورات تمام نشده، عمليات نيمه كاره، فايلهاي باز و … 
تعداد فرودگاهها يا همان برچسبها نبايد زياد (حتي بيش از يكي) باشد، مگر آنكه نظم كاملا مشخصي در آن حكم فرما باشد.
حتي​الامكان به جاي استفاده از پرش، از دستوراتي نظير try..finaly, try..except استفاده كنيد.
فقط زماني از پرش استفاده كنيد كه پرش به وضوح برنامه شما كمك كند. به عبارت ديگر با استفاده از پرش بتوانيد خوانائي برنامه را افزايش دهيد. 
پرش فقط توسط برنامه​نويسان بسيار حرفه​اي و افرادي كه داراي نظم بسيار زياد هستند خوب استفاده مي​شود. اگر اهل نظم و ديسيپلين شديد نيستيد، به هيچ عنوان از پرش استفاده نكنيد.










اگر مي خواهيد بپريد، بايد جاي محكمي براي فرود آمدن داشته باشيد! پريدن بد نيست، اگر بدانيد كجا مي�پريد!

